

Zadanie 15.

Transakcje kasowe

Wykorzystane funkcje: CZĘSTOŚĆ, SUMA.ILOCZYNÓW

Wykorzystane narzędzia: formuły, funkcje

Poziom trudności: 4/5

Treść zadania

W pliku *zadanie_15.xlsx* w arkuszu *dane* znajdują się informacje dotyczące wysokości pojedynczych transakcji kasowych dokonywanych w wybranym dniu w pewnym markecie. Dodatkowo znajdują się tam również informacje o liczbie towarów jednorazowo kupowanych przez klientów.

1. Przygotuj zestawienie tabelaryczne, w którym znajdzie się informacja o liczbie zrealizowanych transakcji, podczas których zakupiono do 5 towarów włącznie, od 6 do 10 włącznie, od 11 do 30 włącznie oraz od 31 do 50 włącznie.
2. Oblicz, ile było transakcji o wysokości do 50 zł włącznie i transakcji w wysokości od 200 do 250 zł włącznie.
3. Oblicz, ile było transakcji o wysokości do 50 zł włącznie i liczbie towarów powyżej 6 sztuk.
4. Oblicz, ile było transakcji o wysokości od 50 zł do 100 zł włącznie lub od 200 zł do 250 zł włącznie i liczbie towarów powyżej 10 sztuk.

Rozwiązanie

Polecenie 1.

Zestawienie, o którym mowa w pierwszym poleceniu, można przygotować co najmniej dwoma sposobami — korzystając z funkcji CZĘSTOŚĆ albo z funkcji SUMA.ILOCZYNÓW.

1. Rozpocznij od utworzenia tabeli przedstawionej na poniższym rysunku.

	E	F
1	Liczba towarów	
2	od	do
3	1	5
4	6	10
5	11	30
6	31	50

2. Do komórki G2 wpisz tekst Liczba transakcji.
3. Do komórki G3 wstaw funkcję SUMA.ILOCZYNÓW. Jako jej argument *tablica1* wpisz formułę $(\$B\$2:\$B\$809 \geq E3) * (\$B\$2:\$B\$809 \leq F3)$. Jest to dość nietypowe zastosowanie funkcji SUMA.ILOCZYNÓW, która w zwykłej formie oblicza sumę iloczynów liczb z tablic podawanych jako jej kolejne argumenty. Tutaj funkcja ta ma tylko jeden argument, będący koniunkcją dwóch warunków. Użycie tej funkcji w takiej postaci oznacza, że sprawdzane są kolejne wiersze w zakresie B2:B809 pod kątem jednoczesnego spełniania dwóch warunków. Pierwszym z warunków jest $\$B\$2:\$B\$809 \geq E3$, a drugim $\$B\$2:\$B\$809 \leq F3$, co oznacza sprawdzenie, czy któraś z wartości w zakresie B2:B809 jest większa bądź równa od wartości podanej w komórce E3 i jednocześnie mniejsza bądź równa od wartości podanej w komórce F3. Spełnienie każdego z tych warunków daje w rezultacie wartość 1, co jest następnie podliczane przez funkcję SUMA.ILOCZYNÓW. Cała formuła w komórce G3 powinna mieć postać:

$=\text{SUMA.ILOCZYNÓW}((\$B\$2:\$B\$809 \geq E3) * (\$B\$2:\$B\$809 \leq F3))$

Formułę z komórki G3 przeciągnij w dół arkusza aż do komórki G6.

	E	F	G	H
1	Liczba towarów			
2	od	do	Liczba transakcji	
3	1	5	263	
4	6	10	241	
5	11	30	233	
6	31	50	71	

Aby przygotować takie samo zestawienie przy pomocy funkcji CZĘSTOŚĆ:

1. Zaznacz komórki z zakresu H3:H6.

	E	F	G	H	I
1	Liczba towarów				
2	od	do	Liczba transakcji		
3	1	5	263		
4	6	10	241		
5	11	30	233		
6	31	50	71		
7					

2. Wstaw funkcję CZĘSTOŚĆ. Pierwszy parametr, czyli *tablica_dane*, to zakres komórek, które mają zostać zliczone. Wpisz B2:B809. Drugi argument to tablica zawierająca przedziały, w których mają być zliczane punkty, do przedostatniej

klasy włącznie. Jako argument *tablica_przedziały* wpisz F3:F5. Funkcja ta wlicza do pierwszego przedziału wszystkie wartości ze zbioru danych, które są mniejsze lub równe od pierwszej górnej granicy przedziału, a do ostatniego — wszystkie wartości większe od górnej granicy przedostatniego przedziału.

3. Ponieważ funkcja CZĘSTOŚĆ jest funkcją tablicową (zwracającą w rezultacie swojego działania więcej niż jedną wartość), zatwierdź ją kombinacją klawiszy *Ctrl+Shift+Enter*.

	E	F	G	H
1	Liczba towarów			
2	od	do	Liczba transakcji	
3	1	5	263	263
4	6	10	241	241
5	11	30	233	233
6	31	50	71	71

Polecenie 2.

Rozwiązania poleceń 2 – 4 wykonaj w nowej kopii arkusza z danymi. Nazwij ten arkusz *rozwiązania 2 – 4*.

Aby obliczyć, ile było transakcji o wysokości do 50 zł włącznie i transakcji o wysokości od 200 do 250 zł włącznie:

1. Do komórki E1 wpisz tekst Liczba transakcji o wysokości do 50 zł włącznie i transakcji o wysokości od 200 do 250 zł włącznie.
2. Do komórki E2 wstaw funkcję SUMA.ILOCZYNÓW. Jako jej argument *tablica1* wpisz formułę $(A2:A809<=50)+(A2:A809>=200)*(A2:A809<=250)$. Działanie tej funkcji jest takie samo jak opisane powyżej, z tą różnicą, że sprawdzana jest alternatywa dwóch warunków (symbol + w formule), z których jeden jest koniunkcją (symbol * w formule) dwóch innych. Sprawdzane jest zatem, czy transakcje należą do przedziału od 200 zł do 250 zł lub do przedziału o górnej granicy 50 zł. Formuła w komórce E2 powinna mieć postać:

$$=SUMA.ILOCZYNÓW((A2:A809<=50)+(A2:A809>=200)*(A2:A809<=250))$$

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	Kwota	Liczba towarów			Liczba transakcji o wysokości do 50 zł włącznie i transakcji o wysokości od 200 do 250 zł włącznie									
2	128,95	9			81									
3	128,02	13												
4	29,72	5												
5	109,11	1												

Polecenie 3.

Aby obliczyć, ile było transakcji o wysokości do 50 zł włącznie i liczbie towarów powyżej 6 sztuk:

1. Do komórki *E4* wpisz tekst Liczba transakcji o wysokości do 50 zł włącznie i liczbie towarów powyżej 6 sztuk.
2. Do komórki *E5* wstaw funkcję SUMA.ILOCZYNÓW. Jako jej argument *tablica1* wpisz formułę $(A2:A809 \leq 50) * (B2:B809 > 6)$. To, co odróżnia tę formułę od poprzednich, to fakt, że przeszukiwane pod kątem jednoczesnego spełnienia dwóch warunków są dwa różne zakresy danych (zakresy o tym samym rozmiarze). Cała formuła w komórce *E5* powinna mieć postać:

$=\text{SUMA.ILOCZYNÓW}((A2:A809 \leq 50) * (B2:B809 > 6))$

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	Kwota	Liczba towarów			Liczba transakcji o wysokości do 50 zł włącznie i transakcji o wysokości od 200 do 250 zł włącznie									
2	128,95	9			81									
3	128,02	13												
4	29,72	5			Liczba transakcji o wysokości do 50 zł włącznie i liczbie towarów powyżej 6 sztuk									
5	109,11	1			13									
6	160,36	13												
7	93,69	12												
8	85,02	5												

Polecenie 4.

Aby obliczyć, ile było transakcji o wysokości od 50 zł do 100 zł włącznie lub od 200 zł do 250 zł włącznie i liczbie towarów powyżej 10 sztuk:

1. Do komórki *E7* wpisz tekst Liczba transakcji o wysokości od 50 zł do 100 zł włącznie lub od 200 zł do 250 zł włącznie i liczbie towarów powyżej 10 sztuk.
2. Do komórki *E8* wstaw funkcję SUMA.ILOCZYNÓW. Jako jej argument *tablica1* wpisz formułę $((A2:A809 \geq 50) * (A2:A809 \leq 100) + (A2:A809 \geq 200) * (A2:A809 \leq 250)) * (B2:B809 > 10)$. W tym przypadku sprawdzana jest prawdziwość koniunkcji dwóch warunków. Pierwszy z warunków służy sprawdzeniu przynależności transakcji do jednego z przedziałów: od 50 zł do 100 zł lub od 200 zł do 250 zł. Drugi natomiast służy sprawdzeniu, czy liczba zakupionych towarów jest większa niż 10. Cała formuła w komórce *E8* powinna mieć postać:

$=\text{SUMA.ILOCZYNÓW}(((A2:A809 \geq 50) * (A2:A809 \leq 100) + (A2:A809 \geq 200) * (A2:A809 \leq 250)) * (B2:B809 > 10))$

